Control With PROC CONTENTS

Gary McQuown, Data and Analytic Solutions Inc. Fairfax, Virginia

ABSTRACT

PROC CONTENTS can be used for many tasks beyond the obvious. From assistance with data manipulation to directing data driven applications, PROC CONTENTS may just be the tool you need.

INTRODUCTION

PROC CONTENTS provides information about the contents of a data set. This information is most often viewed to determine if a variable is present or has a certain attribute. This process can be taken further by using the output from PROC CONTENTS within the data step to select and manipulate variables. The larger the number of variables in the data set, the more advantageous this process becomes.

The output data set from PROC CONTENTS contains 40 variables that describe the variables or the data set. Most can be used to select or manipulate variables. Some of the most beneficial are:

LABEL	Variable Label
LENGTH	Variable Length
LIBNAME	Library Name
MEMNAME	Library Member Name
MODATE	Last Modified Date
NAME	Variable Name
NOBS	Observations in Data Set
VARNUM	Variable Number.

TASK

For every variable in a given data set that starts with the letter "X", replace any blanks that are not in the first position with a dot.

The process is both well defined and repetitive, so it is ideal for automating.

EXAMPLE

```
/* Create test data */
data test;
  X1 = "ab def" ;
  X2 = "a c ef" ;
  Y3 = "John Doe" ;
run;

%macro xblank(ckdsn);
/*************************************
XBLANK.SAS
Purpose: Translate non-first position blanks in "X"
codes to dots
Warning: This program will overwrite the original
file
Parameters:  ckdsn = the name of the input file
*************************************/

/* Proc Contents to produce list of variables */
proc contents data=&ckdsn noprint out=cont ;
run;

/* Proc SQL to create macro list of variables */
proc sql noprint;
  select name as name format=$8.
    into :xcode separated by ' ' from cont
    where (substr(name,1,1) eq "X") ;
quit;

data &ckdsn  ;
 set &ckdsn ;

 %LET i = 1;
 %DO %WHILE ((%scan(&xcode, &i))>' ');
  %let varn  = %scan(&xcode, &i);

   /* data manipulation */
   &varn =   left(substr(&varn,1,1)||
  translate(substr(&varn,2), '.', ' '));

 /* increment the loop */
 %LET i = %EVAL(&i+1);
 %END;
run;

%mend xblank;
%xblank(test);
```

RESULTS

```
Obs     X1       X2      Y3
 1     ab.def   a.c.ef   John Doe
```

CONTACT INFORMATION

Gary McQuown
Data and Analytic Solutions Inc.
10502 Assembly Drive
Fairfax, VA 22030
mcquown@DASconsultants.com